

# Quantitation of Interactive Effects between Anticancer Agents Using SAS® Macros and SAS/GRAPH®

Yalan Wang, The Ohio State University, Columbus, OH 43210

Jun Zuo, Ph.D., STATPROBE, Inc., Ann Arbor, MI 48108

M. Guill. Wientjes, Ph.D., The Ohio State University, Columbus, OH 43210

## ABSTRACT

The interactive effects of two anticancer agents are frequently evaluated using three different types of graphs: concentration-effect curves, isobolograms, or combination index curves. In order to determine the cytotoxicity of the combined agents at an array of combinations, the experimental cytotoxicity data are usually processed by using statistical software and spreadsheet software separately. In general, the use of separate software applications is time-consuming and introduces the risk of errors made during data transfer. To avoid these pitfalls, we developed a SAS based system that processes the raw cytotoxicity assay data and generates three types of graphs and related parameters. The system consists of a series of SAS macros that can be structured in three levels (menu program, main flow macro, and individual function macros) and classified into three categories (data manipulation, model estimation, and graph presentation). Three curves are generated using SAS/GRAPH. It is efficient and convenient for researchers to this system to measure the interactive effects of combinations of two anticancer agents.

## INTRODUCTION

Combination of two or more anticancer agents is believed to be essential to achieving a cure for most drug-sensitive cancers. For this and other reasons, scientists from various disciplines including pharmacology, toxicology, statistics, mathematics, and epidemiology, have been in search for synergistic drug combinations for many years. Among those methods for assessing drug interactions (see Greco et al. (1995)), three approaches are most commonly used: the Sigmoid-Emax equation (Holford and Sheiner, 1981), the equation of Loewe additivity (Loewe and Muischnek, 1926), and the median-effect equation (Chou and Talalay, 1984). Each equation can be applied with experimental data to generate its corresponding graphs used as a visual tool to assess the interactive effects of two-drug combinations. The corresponding graphs associated with three equations are the concentration-effect curves, isobologram curves, and combination index curves, respectively.

Software to evaluate combination data with the median-effect equation is commercially available. However, to our knowledge, no single software is available to generate all three types of graphs. Therefore, it is desirable to develop a system which can process the raw experimental data and produce those graphs and related parameters.

In this paper, the authors present a SAS based system that can be used to process the raw cytotoxicity data of two anticancer agents and to generate three types of graphs and related parameters. The system consists of a series of SAS macros that can be structured in three levels: menu program, main flow macro, and individual function macros, and classified into three categories: data manipulation, model estimation, and graph presentation. The graphs are presented using SAS/GRAPH software. The automated output from colorimetric readings of 96-well plates is used as input for the SAS program.

In the next section, three approaches for assessing interactive effects of combinations are briefly discussed with three corresponding curves. The major SAS macros are then

demonstrated in three different levels: menu program, main flow macro, and individual function macros. An example is given to measure interactive effects between two anticancer agents. In the conclusion section, the further enhancement of system is briefly discussed.

## MEASURING INTERACTIVE EFFECTS

When two anticancer drugs interact, the cytotoxicity of the combination can be either enhanced or diminished. The former is synergy, and the latter is called antagonism. Because synergism (and antagonism) is commonly defined as a greater (or lesser) pharmacological effect for a two-drug combination than what would be predicted for "no interaction" (null reference model), their definitions critically depend on the reference model for "interaction". No consensus exists relative to this issue, and as a consequence, several equally valid models for evaluation of interaction exist. The three graphical approaches accommodated in this SAS program are

- Concentration-effect curve
- Isobologram curve
- Combination index curve.

### Concentration-Effect Curve

The first approach is based on the Sigmoid-Emax model, a common-used name for the Hill model (Hill, 1910). The Sigmoid-Emax model can be represented by the following equation for an inhibitory drug:

$$(1) \quad E = E_{\max} \cdot (D/IC_{50})^m / (1 + (D/IC_{50})^m)$$

Where E is the measured effect (response), such as percentage of cell surviving; D is the concentration of the drug;  $E_{\max}$  is the full range of inhibitory effect by the drug, often 100%;  $IC_{50}$  is the median effective concentration of drug; and  $m$  is a curvature parameter. Experimentally, combinations of drug exposure concentration (D, normalized as  $D/IC_{50}$ ) and cytotoxic effects (E) are determined. The fitting parameters,  $E_{\max}$ ,  $IC_{50}$ , and  $m$  can be estimated by nonlinear regression using PROC NLIN in SAS/STAT. The concentration-effect curve (see Figure 1 in Appendix) can be plotted by connecting the predicted values from the constructed model.

Based on the concentration-effect model, a fitted  $IC_{50}$  of each individual drug can be obtained and used to normalize the concentration of each drug. The combination concentration of a mixture of two drugs (with ratio A/B) can then be calculated as:

$$(2) \quad \text{Combination Concentration} = A \cdot D_A / IC_{50A} + B \cdot D_B / IC_{50B}$$

and is expressed in unit of  $IC_{50}$  equivalent.

Using the normalized concentration, multiple concentration-effect curves for two-drug combinations and two single agents can be estimated and constructed. By definition, single agents have an  $IC_{50} = 1.0$ . (see Figure 2 in Appendix). A leftward shift of the cytotoxicity curve of a drug combination ( $IC_{50} < 1.0$ ) indicates synergy of the drug combination. Experimentally, the concentration ratios between two drugs are kept constant for each combination curve.

## Isobologram Curve

Isobologram analysis of drug synergy concentrates on changes in the combination concentrations required to achieve a defined level of effect, commonly 50% of maximal cytotoxicity. This approach uses the equation of Loewe additivity:

$$(3) \quad IC_{50,A \text{ and } B} = D_A/IC_{50,A} + D_B/IC_{50,B}$$

where  $IC_{50,A \text{ and } B}$  is the combination concentration of drug A and B at 50% inhibition,  $IC_{50,A}$  and  $IC_{50,B}$  are the concentrations of drugs to result in 50% inhibition for each respective drug alone, and  $D_A$  and  $D_B$  are concentrations of each drug in the mixture that yield 50% inhibition. The isobologram curve (see Figure 3 (1) in Appendix) is generated by plotting doses of drug A vs. doses of drug B which are predicted in combination to achieve a set endpoint, such as 50% cell growth inhibition. A standard line of Loewe additivity is included to indicate a lack of interaction. Points below the line indicate synergy, while points above the line indicate antagonism.

## Combination Index Curve

Because the isobologram lacks a summary measurement of the intensity of interaction, the third type of graph, combination index curve, can be derived based on the medium-effect equation developed by Chou and Talalay (1984):

$$(4) \quad CI = D_A/IC_{X,A} + D_B/IC_{X,B}$$

where CI is combination index,  $IC_{X,A}$  and  $IC_{X,B}$  are the concentrations of drugs to result in X% inhibition for each respective drug alone, and  $D_A$  and  $D_B$  are concentrations of each drug in the mixture that yield X% inhibition. The combination index curve or modified isobologram (see Figure 3 (2) in Appendix) is generated by plotting CI vs X, ranging from 0% to 100%. The graph allows easy quantitation of drug interaction over the full range of drug effect. The drug interaction is readily identified at any level of inhibition. The assumption used in the model is assuming that the effects of the two drugs are nonmutually exclusive.

## DEVELOPING SAS MACROS

Using base SAS, SAS/STAT, and SAS/GRAPH, the customized SAS macros are developed to efficiently and effectively capture the above theoretical equations, to process raw experiment data, and to present flexible graphs. These SAS macros are formed as a system with three levels: menu program, main flow macro, and individual function macros.

### Menu Program

The menu program works as an interactive tool in which researchers can input or change environment parameters and raw data set parameters. The environment parameters contain 1) paths to input raw data, to output SAS or text files, and to output graphs in different ways; 2) destinations of graphs and parameters of destinations; 3) paths and names of individual function macros. The parameters related to raw data set include number of combinations, ratios of combinations, names of combinations, variable names in data set, and others. The menu program, which was used to deal with one raw data sets, is shown as follows:

```
*** Set up paths to include macros, input raw data, and output SAS files,
text files, or graph files ***;
%LET MACROS=c:\data\synergy\macro;
%LET DATAPATH=c:\data\synergy\data;
%LET OUTFILE=c:\data\synergy\output;
%LET GRPFIL=c:\data\synergy\graph;

*** Set up computer output parameters for graphs ***;
%LET PRINTER=hp1j4si;
%LET SCREEN=win
```

```
%LET WORDFILE=cgmmw6c;

*** Choose current destination of graphs: PRN - printer, SCR - screen,
WPF -- Wordperfect file, WDF -- Word file ***;
%LET DEST=PRN;

*** Option for data source listed in graph ***;
%LET SOURCE=NO;

*** Include function macros ***;
%INCLUDE "&MACROS.UTILITY.MAC";
%INCLUDE "&MACROS.PRODATA.MAC";
%INCLUDE "&MACROS.EMAX.MAC";
%INCLUDE "&MACROS.ICVALUE.MAC";
%INCLUDE "&MACROS.IPRICVAL.MAC";
%INCLUDE "&MACROS.CALCIC50.MAC";
%INCLUDE "&MACROS.CALCISO.MAC";
%INCLUDE "&MACROS.CALCCI.MAC";
%INCLUDE "&MACROS.SETGRAPH.MAC";
%INCLUDE "&MACROS.SEPLLOT1.MAC";
%INCLUDE "&MACROS.SEPLLOT6.MAC";
%INCLUDE "&MACROS.SEPLLOT6S.MAC";
%INCLUDE "&MACROS.ISOPLLOT.MAC";
%INCLUDE "&MACROS.CIPILOT.MAC";
%INCLUDE "&MACROS.REPLAY2.MAC";
%INCLUDE "&MACROS.MAIN.MAC";

*** Assign number of combinations:          drgnum ***;
***      number of groups in each input file:  rpnum ***;
***      number of obs in each group:         ordnum ***;
%LET DRGNUM = 6;
%LET GRPNUM = 1;
%LET ORDNUM = 10;

*** Indicate variable list in original raw data ***;
%LET RAWVARS=DRUG $6. GROUP ORDER CONC
X1 X2 X3 X4 X5 XC;

*** Assign ratio values for combinations ***;
%LET P1 = 1.0;
%LET P2 = 0.8;
%LET P3 = 0.6;
%LET P4 = 0.4;
%LET P5 = 0.2;
%LET P6 = 0.0;

*** Assign two drug agent names and titles ***;
%LET DRUG1 =DRUGA;
%LET DRUG2 =DRUGB;
%LET TITL1 =DRUG A;
%LET TITL2 =DRUG B;

*** Assign combination names corresponding ratio values ***;
%LET DRGP1 =DRUGA;
%LET DRGP2 =A08B02;
%LET DRGP3 =A06B04;
%LET DRGP4 =A04B06;
%LET DRGP5 =A02B08;
%LET DRGP6 =DRUGB;

*** Assign raw data file names to macro variables ***;
%LET INDAT1=sugi22.txt;

*** Link to temp names to raw data files (text files) ***;
FILENAME IN1 "&DATAPATH.&INDAT1";

*** Call main analysis macro ***;
%MAIN(INFILE=IN1, FILENAME=&INDAT1);
```

### Diagram 1 Menu Program

### Main Flow Macro

The purpose of main flow macro is to call individual function macros to conduct data manipulation, model estimation, and graph presentation. The flow macro is written with comments or notes as follows:

```
%MACRO MAIN(INFILE=, FILENAME=,OUTGRAPH=YES);
*** Read in raw data (ASCII file) ***;
DATA RAWDATA;
INFILE &INFILE;
```

```

INPUT &RAWVARS;
RUN;

%SORT(DATA=RAWDATA, SORTVARS=DRUG GROUP ORDER);

*** Print raw data ***;
%PRIDATA(INDS=RAWDATA, VARS=DRUG CONC X1 X2 X3 X4 X5
XC, TITLE2=RAW DATA FROM SRB ASSAY);

*** Process raw data to obtain two datasets: RAW1: containing
inhibition and concentration data for two principal drugs and
RAW2: containing inhibition for principal and combination drugs***;
%PRODATA(RAWDATA, RAW1, RAW2);

*** Estimate EMAX model for two principal drugs using emax
macro and obtain RES1: containing estimated parameters and
predicted values ***;
%EMAX(RAW1, RES1);

*** Obtain ic-values for two drugs ***;
%ICVALUE(RES1, RESULT1);

*** Print ic-values and estimated parameters ***;
%PRIICVAL(INDS=RESULT1, TITLE2=EFFECTS OF TWO
PRINCIPAL DRUGS: &TITL1 AND &TITL2);

*** Merge ic-values with RAW1 ***;
%SORT(DATA=RESULT1, SORTVARS=DRUG GROUP);
DATA RAW1;
MERGE RAW1 RESULT1;
BY DRUG GROUP;
RUN;

*** Calculate IC50 equivalent concentrations for all drugs ***;
%CALCIC50(RAW1, IC50EQ);

*** Merge created concentration data with RAW2 ***;
%SORT(DATA=IC50EQ, SORTVARS=DRUG GROUP ORDER);
%SORT(DATA=RAW2, SORTVARS=DRUG GROUP ORDER);
DATA INHDATA;
MERGE RAW2 IC50EQ;
BY DRUG GROUP ORDER;
KEEP DRUG GROUP ORDER CONC INH STD;
RUN;

*** Print data with ic equivalent concentrations ***;
%PRIDATA(INDS=INHDATA, VARS=DRUG CONC INH STD,
TITLE2=INHIBITION DATA WITH IC50 EQUIVALENT
CONCENTRATIONS);

*** Estimate EMAX model for all combination drugs using
emax macro and obtain RES2: containing estimated
parameters and predicted values ***;
%EMAX(INHDATA, RES2);

*** Obtain ic-values for all drugs ***;
%ICVALUE(RES2, RESULT2);

*** Print ic-values and estimated parameters ***;
%PRIICVAL(INDS=RESULT2, TITLE2=EFFECTS OF
COMBINATIONS OF &TITL1 AND &TITL2);

*** Calculate isobologram data ***;
%CALCISO;

*** Output isobologram data ***;
%PRIDATA(INDS=ISODATA, VARS=DRUG &DRUG1
&DRUG2, TITLE2=ISOBOLOGRAM DATA);

*** Calculate combination index (CI) ***;
%CALCCI;

*** Output combination index ***;
%PRIDATA(INDS=CIDATA, VARS=ICID %DO I = 1 %TO &DRGNUM;
&&DRGP&I %END;, TITLE2=COMBINATION INDEX);

*** Generate graphs ***;
%IF %UPCASE(&OUTGRAPH)=YES %THEN %DO;

%SETGRAPH(DEST=INIT);
*** plot graph of survival data for the first drug ***;
%PLOT2(INDS=RES1, drugname=&drug1,
DRUGTITL=&TITL1, gphnum=1, group=1);

```

```

*** plot graph of survival data for the second drug ***;
%PLOT2(INDS=RES1, drugname=&drug2,
DRUGTITL=&TITL2, gphnum=2, group=1);
%SETGRAPH(DEST=&DEST, GRPNAME=FIG1);
%REPLAY2(IGOUT=GRAFCAT);

%SETGRAPH(DEST=INIT);
*** plot graph of survival data ***;
%PLOT6(INDS=RES2, gphnum=1, group=1);
*** plot graph of survival data with standard deviations ***;
%PLOT12(INDS=RES2, gphnum=2, group=1);
%SETGRAPH(DEST=&DEST, GRPNAME=FIG2);
%REPLAY2(IGOUT=GRAFCAT);

%SETGRAPH(DEST=INIT);
*** plot graph of isobologram ***;
%ISOPLOT(INDS=ISODATA2, gphnum=1, group=1);
*** plot graph of modified isobologram ***;
%CIPILOT(INDS=CIDATA, gphnum=2, group=1);
%SETGRAPH(DEST=&DEST, GRPNAME=FIG3);
%REPLAY2(IGOUT=GRAFCAT);
QUIT;
%END;

%CLEANUP(RAWDATA RAW1 RAW2 RES1 RES2 RESULT1
RESULT2 IC50EQ INHDATA ISODATA CIDATA);
%MEND MAIN;

```

## Diagram 2 Main Flow Macro

### Individual Function Macros

All individual function macros can be found in the main flow macro and classified into three categories: data manipulation, model estimation, and graph presentation.

#### Macros for Data Manipulation

The following SAS macros are used to manipulate data sets:

- %PRODATA -- to process and separate raw data into two datasets: one containing data for two principal drugs, the other including inhibition variable for combinations.
- %PRIDATA -- to output data to SAS output window and a disk file.
- %PRIICVAL -- to output ic values to SAS output window and a disk file.
- %SORT -- to sort a SAS data set (included in %UTILITY).
- %CLEANUP -- to clean up SAS data set(s) (included in %UTILITY).

#### Macros for Model Estimation

These macros are used to estimate Equation (1) or calculate values based on Equations (1) to (4). The macros include:

- %EMAX -- to estimate Equation (1) using PROC NLIN.
- %ICVALUE -- to calculate ic values for two principal drugs based on Equation (1).
- %CALCIC50 -- to calculate ic50 values for combinations based on Equation (2).
- %CALCISO -- to calculate isobologram data based on Equation (3).
- %CALCCI -- to calculate combination index based on Equation (4).

The %EMAX macro is displayed as follows.

```

%MACRO EMAX(INDS, OUTDS);

*** take log to concentrations ***;
DATA RAW;
SET &INDS;
IF CONC>0 THEN LOGCONC=LOG10(CONC);
ELSE LOGCONC=.;
RUN;

%SORT(DATA=RAW, SORTVARS=DRUG GROUP CONC);

*** extract drug name and group number ***;
DATA TITLES;

```

```

SET RAW;
BY DRUG GROUP;
IF FIRST.GROUP;
KEEP DRUG GROUP ;
RUN;

%SORT(DATA=TITLES,SORTVARS=DRUG GROUP);

*** create a new data with log concentrations for each drug
and group ***;
DATA NEW; SET TITLES;
DO LOGCONC=-4 TO 3 BY 0.1;
CONC=10**LOGCONC;
INH=.;
OUTPUT;
END;
RUN;

%SORT(DATA=NEW,SORTVARS=DRUG GROUP CONC);

*** merge new data with original data by drug concentration
and group ***;
DATA IC50;
MERGE NEW RAW;
BY DRUG CONC GROUP;
RUN;

%SORT(DATA=IC50,SORTVARS=DRUG GROUP CONC);

*** apply nonlinear EMAX model ***;
PROC NLIN BEST=3 MAXITER=100
METHOD=MARQUARDT NOPRINT;
BY GROUP DRUG;
PARAMETER EMAX=20,50,100
N=1,3 K=0.1,1,10,100,500;
BOUNDS 0.5<N<5,K>0.001;
IF CONC>0 THEN DO;
MODEL INH=EMAX*(1-CONC**N/(K**N+CONC**N));
DER.EMAX=1-(CONC**N)/(K**N+CONC**N);
FACT1=1/(K**N+CONC**N)*LOG(CONC)*CONC**N;
FACT2=CONC**N*((1/(K**N+CONC**N))**2)*
(LOG(K)*K**N+LOG(CONC)*CONC**N);
DER.N=EMAX*(FACT1-FACT2);
DER.K=EMAX*(CONC**N)/((K**N+CONC**N)**2)*
N*K**(N-1);
END;
IF CONC=0 THEN DO;
MODEL INH=EMAX;
DER.EMAX=1-(CONC**N)/(K**N+CONC**N);
DER.N=0;
DER.K=EMAX*(CONC**N)/((K**N+CONC**N)**2)*
N*K**(N-1);
END;
_WEIGHT_=1;
OUTPUT OUT=&OUTDS P=INHAT R=RESIDUAL
PARMS=EMAX N K;
RUN;

%SORT(DATA=&OUTDS,SORTVARS=DRUG GROUP);

%CLEANUP(RAW NEW TITLES IC50);
%MEND EMAX;

```

**Diagram 3 Macro to Estimate Sigmoid-Emax Equation**

#### Macros for Graph Presentation

In order to output graphs into a selected destination, a graphic environment macro is created (%SETGRAPH). Other graph presentation macros include:

- %SEPLOT1 -- to plot a concentration-effect curve (Sigmoid-Emax curve) for a single drug.
- %SEPLOT6 -- to plot multiple concentration-effect curves without standard deviation for combinations of drugs.
- %SEPLOT6S -- to plot multiple concentration-effect curves with standard deviation for combinations of drugs.
- %ISOPLOT -- to plot an isobologram curve
- %CILOT -- to plot a combination index curve.
- %REPLAY2 -- to replay or put two plots into one page.

Two of the above macros, %setgraph and %replay2, are displayed as follows.

```

%MACRO SETGRAPH(DEST=,GRPNAME=);

%If &grpname ^= %then %do;
  Filename cgmout "&grpname.\&grpname.cgm";
%End;

goptions reset=all nodisplay;

%If %upcase(&dest)=INIT %then %do;
  goptions device=&screen reset=global gunit=pct
  gsfmode=replace hsize=5.5 in vsize=4 in border
  hpos=70 vpos=60 htext=1.25 cells ftext=swiss;
%Else %do;
  *** goptions to send the plot directly to the printer ***;
  %If %upcase(&dest) =PRN %then %do;
    goptions gaccess='SASGASTD>LPT1:' device=&printer
    hpos=80 vpos=70 nopolygonfill nofill
    rotate=landscape ftext=swiss htext=1.25 cells
    hsize=9 in vsize=6 in horigin=1 in vorigin=0.75 in;
  %End;

  *** option to route the plot to Word processor file ****;
  %If %upcase(&dest) =WPF %then %do;
    goptions norotate device=wordfile gsfname=cgmout
    gsfmode=replace hpos=80 vpos=70 htext=1.25 cells
    ftext=swiss;
  %End;

  *** goptions to view the plot on screen for specific and
  default purposes ***;
  %If %upcase(&dest) =SCR or
  (%upcase(&dest) ^=PRN and %upcase(&dest) ^=WPF)
  %then %do;
    goptions device=&screen reset=global gunit=pct
    ftext=swiss htitle=6 htext=1.25;
  %End;
%End;
%MEND SETGRAPH;

```

**Diagram 4 Macro to Set Up Environment for Graph Presentation**

```

%MACRO REPLAY2(IGOUT=);
GOPTIONS DISPLAY;

PROC GREPLAY IGOUT=&IGOUT
TC=TEMPCAT NOFS;
TDEF NEWTWO DES='TWO GRAPHS IN ONE PAGE'
1/LLX=2 LLY=2 ULX=2 ULY=98
URX=48 URY=98 LRX=48 LRY=2
2/LLX=52 LLY=2 ULX=52 ULY=98
URX=98 URY=98 LRX=98 LRY=2;
TEMPLATE NEWTWO;
TREPLAY 1:GPLOT
2:GPLOT1;
DELETE _ALL_;

RUN;
QUIT;

PROC DATASETS LIBRARY=WORK
MEMTYPE=CATALOG;
DELETE &IGOUT GSEG;
QUIT;
%MEND REPLAY2;

```

**Diagram 5 Macro to Replay Two Graphs in One Page**

#### EXAMPLE: Interactive Effect of Combinations of Drug A and Drug B

Based on different action mechanisms and toxicity of Drug A and Drug B, the nature and magnitude of the interaction between the agents are investigated in vitro by a cytotoxicity assay. Both the individual agents and their four alternative ratios of combinations (Drug A : Drug B = 0.8:0.2, 0.6:0.4, 0.4:0.6, 0.2:0.8) were included to assess their interaction.

Using the SAS based system, the concentration-effect curves of Drug A and Drug B are plotted and shown in Appendix (Figure 1). After normalization of concentration using  $IC_{50}$  value, the concentration-effect curves for all combinations and single agents can be estimated and plotted in one scale with  $IC_{50}$  equivalent as x-axes and percentage survival as y-axes (see Figure 2 in Appendix). The nature and intensity of interaction can be visualized through the plot at 50% growth inhibition level. If the curve shifts to the left of the cross points of the two single agents, synergy is indicated. In this example, all the four combination curves shift to the left of the single agents, which shows significant synergistic interaction of Drug A and Drug B. Figure 2(1) in Appendix is ideal curves, and Figure 2 (2) are the fitted curves by experimental data with standard deviation.

The results from four combinations at 50% inhibition level in this plots can also be utilized to construct isobologram curve (Figure 3 (1) in Appendix) for the purpose of easier and more direct identification of interaction. Furthermore, the combination-index curves (Figure 3 (2) in Appendix) derived from medium-effect methods were generated to give a summary of interactive effect along the growth inhibition level 10% to 90%. From the plot, it is obvious that the combination of Drug A : Drug B = 0.8:0.2 and 0.6:0.4 shows synergy at all inhibition level and the magnitude of interaction is approximately twofold of synergy. The ratio of Drug A to Drug B = 0.4:0.6 and 0.2:0.8 is synergistic from 10% inhibition level to 70% inhibition level, additive at 80% inhibition level, but antagonistic at 90% inhibition level.

The SAS based system has been used in our Lab to efficiently and conveniently process cytotoxicity data for measuring the possible synergistic effects of combinations for many different anticancer agents, such as Taxol, geldanamycin, suramin, and cytochalasin E.

## CONCLUSION

A SAS based system is developed in this paper in order to generate three types of analysis graphs (concentration-effect curve, isobologram, or combination index curve) for measuring interactive effects of combinations of two anticancer drugs. It is efficient and convenient for researchers to use this system to conduct assessment of anticancer agent interaction.

Further enhancement is under consideration by integrating the current system with SAS/AF<sup>®</sup> software in order to create a user-friendly interface. Specifically, the menu program will be assembled into various SAS/AF FRAME entries and the main flow macro will be replaced using corresponding SCL code.

## REFERENCES

- Chou, T.C., and Talalay, P.: Quantitative analysis of dose-effect relationships: the combined effects of multiple drugs or enzyme inhibitors. *Adv. Enzyme Regul.* 22: 27-55, 1984.
- Greco, W. R., Bravo, G., and Parsons, J.C.: The search for synergy: a critical review from a response surface perspective. *Pharmacological Reviews* 47: 331-385, 1995.
- Hill, A. V.: The possible effects of the aggregation of the molecules of haemoglobin on its dissociation curves. *J. Physiol.* 40: iv-vii, 1910.
- Holford, N. H. G., and Scheiner, L. B.: Understanding the dose-effect relationship: clinical applications of pharmacokinetic-pharmacodynamic models. *Clin. Pharmacokinet.* 6: 429-453, 1981.
- Loewe, S., and Muischnek, H.: Effect of combinations: mathematical basis of problem. *Arch. Exp. Pathol. Pharmacol.* 114: 313-326, 1926.

SAS Institute Inc.: SAS/GRAPH Software: Reference (Volume 1 and Volume 2), Cary, NC: SAS Institute Inc., 1990.

SAS Institute Inc.: SAS/STAT User's Guide (Volume 2), Cary, NC: SAS Institute Inc., 1990.

## ACKNOWLEDGEMENT

SAS, SAS/GRAPH, SAS/STAT, and SAS/AF are the registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. Other brands and product name are registered trademarks or trademarks of their respective companies. ® indicates USA registration.

## AUTHORS' ADDRESS


Yalan Wang, M.S.  
College of Pharmacy  
The Ohio State University  
500 W. 12th Ave.  
Columbus, Ohio 43210  
Phone: (614)-293-3267  
E-mail: Wang.253@postbox.acs.ohio-state.edu

Jun Zuo, Ph.D.  
STATPROBE, Inc.  
3885 Research Park  
Ann Arbor, Michigan 48108  
Phone: (313)769-5000 x 139  
Email: jzuo@ix.netcom.com


M. Guill Wientjes, Ph.D.  
College of Pharmacy  
The Ohio State University  
500 W. 12th Ave.  
Columbus, Ohio 43210  
Phone: (614)292-4055  
E-mail: Wientjes.1@postbox.acs.ohio-state.edu

**APPENDIX**


**Concentration-Effect Curves, Isobologram, and Combination Index Curve: Drug A and Drug B**



**Figure 1. Concentration-Effect Curves of Two Single Agents**



**Figure 2. Concentration-Effect Curves of Combinations**



**Figure 3. Isobologram and Combination Index Curves**